Some Trapezoid Problems



1 A good problem

Rob Kremer, a math major and teaching Explorer, brought a good problem about trapezoids to an Explorer meeting. There was a joint effort in setting it up and deciding on a method of solution. Rob came up with the best solution, which involved using trigonometry.

1.1 Trapezoid Problem

The trapezoid ACDE shown below has an inscribed equilateral triangle EBD. If AE = 8 and CD = 11, what is the area of the triangle?



1.1.1 drawing

```
> A:= [0,0]:  B := [5,0]:    C:=[7,0]:  De:= [7,2]:    E:= [0,1]:
> eps := .1:
> trap := plottools[polygon]([A,C,De,E],color=yellow):
> tri := plottools[polygon]([E,B,De],color=turquoise):
> labs := MCtools[PT](A-3*[0,eps],"A"), MCtools[PT](B-3*[0,eps],"B"),
  MCtools[PT](C-3*[0,eps],"C"),
  MCtools[PT](De+3*[eps,eps],"D"),
  MCtools[PT](E+3*[-eps,eps],"E"):
```

```
> plots[display](labs,tri,trap);
```



A Solution which uses trig (Kremer): Let x be the side of the triangle and alpha be the measure (in radians) of the angle ABE. Then $\frac{8}{x} = \sin(\alpha)$. Also, since angle EBD is a 60 degree = 2/3* π angle we have angle DBC has measure $\pi - \frac{\pi}{3} - \alpha = \frac{2\pi}{3} - \alpha$. So another equation involving x and alpha is $\frac{11}{x} = \sin(\frac{2\pi}{3} - \alpha)$. Solving each equation for x and setting these equal we get $11\sin(\alpha) = 8\sin(\frac{2\pi}{3} - \alpha)$. Let's use Maple to calculate.

```
> x := 'x':  alpha:='alpha':
> eqn :=expand(11*sin(alpha) = 8*sin(2/3*Pi-alpha));
```

$$eqn := 11\sin(\alpha) = 4\sqrt{3}\cos(\alpha) + 4\sin(\alpha)$$

```
> eqn :=eqn - (4*sin(alpha)=4*sin(alpha));
```

$$eqn := 7\sin(\alpha) = 4\sqrt{3}\cos(\alpha)$$

```
> eqn:=subs(cos(alpha)=sqrt(1-(sin(alpha))^2),eqn);
```

$$eqn := 7\sin(\alpha) = 4\sqrt{3}\sqrt{1-\sin(\alpha)^2}$$

```
> eqn:= lhs(eqn)^2=rhs(eqn)^2;
```

$$eqn := 49\sin(\alpha)^2 = 48 - 48\sin(\alpha)^2$$

```
> eqn :=eqn + (48*(sin(alpha))^2=48*(sin(alpha))^2);
```

$$eqn := 97\sin(\alpha)^2 = 48$$

```
> eqn:= eqn/97;
```

$$eqn := \sin(\alpha)^2 = \frac{48}{97}$$

So, we get the sine of the angle

```
> sola:= sin(alpha)=sqrt(48/97);
```

$$sola := \sin(\alpha) = \frac{4\sqrt{291}}{97}$$

And from that the side of the triangle

```
> solx := x = 8/rhs(sola);
```

$$solx := x = \frac{2\sqrt{291}}{3}$$

So the area is $\frac{\sqrt{3}x^2}{4} = \frac{\sqrt{3}\,291}{9} = 56.00297612$

```
> sqrt(3.)/9*291.;
```

56.00297612

Note angle $\alpha = \arcsin\left(\frac{4\,291^{(\frac{1}{2})}}{97}\right)$

1.2 visual check.

We can check this by making a scale drawing of the trapezoid with it's inscribed equilateral triangle. So, $AB = x\cos(\alpha)$, $AE = x\sin(\alpha)$, $BC = x\cos(\frac{4\pi}{3} - \alpha)$, and $CD = x\sin(\frac{4\pi}{3} - \alpha)$.



1.2.1 drawing

```
> x := 2/3.*291^(1/2):  alpha:= arcsin(4/97.*291^(1/2)):
> A:= [0,0]:  B := [x*cos(alpha),0]:
  C:=[x*cos(alpha)+x*cos(2/3*Pi-alpha),0]:
  De:= [x*cos(alpha)+x*cos(2/3*Pi-alpha),11]:    E:= [0,8]:
> eps := .1:
> trap := plottools[polygon]([A,C,De,E],color=yellow):
> tri := plottools[polygon]([E,B,De],color=turquoise):
> labs := PT(A-3*[0,eps],"A"),PT(B-3*[0,eps],"B"),PT(C-3*[0,eps],"C"),
  PT(De+3*[eps,eps],"D"),PT(E+3*[-eps,eps],"E"):
> plots[display](labs,tri,trap,scaling=constrained);
```



clearly BDE is an equilateral triangle

2 parameterized version of the problem.

This problem has two obvious parameters: a=AE and b=CD, the heights of the trapezoid. We can go back through the solution and get the values of x and alpha in terms of a and b. Then we can investigate the behaviour of the solution as we vary the parameters.

The calculation and drawing sections below were obtained by modifying the calculation and drawing sections above.

2.1 the general calculations

First, assign each variable its own name, thus flushing any previously assigned values. (Note: the restart word does this also.

```
> x := 'x':  alpha:='alpha':  a  := 'a':  b := 'b':
```

Then replace occurences of 8 with a and occurences of 11 with b in the previous equations.

```
> eqn :=b*sin(alpha) = a*sin(2/3*Pi-alpha);
```

$$eqn := b\sin(\alpha) = a\sin(\frac{\pi}{3}+\alpha)$$

We are going to solve this equation for sin(alpha). First, expand the right hand side using the addition formula for sin

```
> eqn :=expand(eqn);
```

$$eqn := b\sin(\alpha) = \frac{1}{2}\,a\,\sqrt{3}\cos(\alpha) + \frac{1}{2}\,a\sin(\alpha)$$

Now, subtract a/2*sin(alpha) from both sides of the equation. This is done by subtracting an equation from eqn.

```
> eqn :=eqn - (a/2*sin(alpha)=a/2*sin(alpha));
```

$$eqn := b\sin(\alpha) - \frac{1}{2}\,a\sin(\alpha) = \frac{1}{2}\,a\,\sqrt{3}\cos(\alpha)$$

We can factor the equation. The maple word factor has been trained to work on equations.

```
> eqn :=factor(eqn);
```

$$eqn := -\frac{1}{2}\sin(\alpha)\,(-2\,b+a) = \frac{1}{2}\,a\,\sqrt{3}\cos(\alpha)$$

Now substitute cos(alpha)=sqrt(1-(sin(alpha))ˆ2) by using the maple word subs.

```
> eqn:=subs(cos(alpha)=sqrt(1-(sin(alpha))^2),eqn);
```

$$eqn := -\frac{1}{2}\sin(\alpha)\,(-2\,b+a) = \frac{1}{2}\,a\,\sqrt{3}\,\sqrt{1-\sin(\alpha)^2}$$

Square both sides to remove the radical. Note: maple doesn't square an equation. But you can square the left hand side (lhs) and the right hand side (rhs) separately and make a new eqn.

```
> eqn:= lhs(eqn)^2=rhs(eqn)^2;
```

$$eqn := \frac{1}{4}\sin(\alpha)^2\,(-2\,b+a)^2 = \frac{3}{4}\,a^2\,(1-\sin(\alpha)^2)$$

Now, to solve for (sin(alpha))ˆ2, expand the right hand side

```
> eqn:=lhs(eqn)= expand(rhs(eqn));
```

$$eqn := \frac{1}{4}\sin(\alpha)^2\,(-2\,b+a)^2 = \frac{3\,a^2}{4} - \frac{3}{4}\,a^2\sin(\alpha)^2$$

and add 3/4*aˆ2*(sin(alpha))ˆ2 to both sides.

```
> eqn :=eqn + (3/4*a^2*(sin(alpha))^2=3/4*a^2*(sin(alpha))^2);
```

$$eqn := \frac{1}{4}\sin(\alpha)^2\,(-2\,b+a)^2+\frac{3}{4}\,a^2\sin(\alpha)^2=\frac{3\,a^2}{4}$$

factor

```
> eqn:=factor(eqn);
```

$$eqn := \sin(\alpha)^2\,(b^2-b\,a+a^2)=\frac{3\,a^2}{4}$$

and divide the right hand side by the second factor of the left hand side. (Note: op(1,x*y) is x and op(2,x*y) is y).

```
> eqn:=op(1,lhs(eqn))=rhs(eqn)/op(2,lhs(eqn));
```

$$eqn := \sin(\alpha)^2=\frac{3\,a^2}{4\,(b^2-b\,a+a^2)}$$

So, we get the sine of the angle α (clearly we want the postive sin since alpha is between 0 and Pi/2)

```
> sola:= sin(alpha)=sqrt(rhs(eqn));
```

$$sola := \sin(\alpha)=\frac{\sqrt{3}\sqrt{\frac{a^2}{b^2-b\,a+a^2}}}{2}$$

And from that the side of the triangle x

```
> solx := x = a/rhs(sola);
```

$$solx := x=\frac{2\,a\,\sqrt{3}}{3\sqrt{\frac{a^2}{b^2-b\,a+a^2}}}$$

So the area is

```
> area = sqrt(3)/2*rhs(solx)^2;
```

$$area=\frac{2\,\sqrt{3}\,(b^2-b\,a+a^2)}{3}$$

At this point, we have the angle alpha, the side x of the inscribed equateral triangle, and the area of the triangle expressed in terms of the parameters a and b, the heights of the trapezoid.

We can make a parameterized diagram to go with this.

2.1.1 parameterized drawing

The process of turning a drawing into a parameterized drawing can be done in more than one way. Here is one. 1. Copy the drawing section and connect the input cells with F4 if need be. 2. Decide on a name for our procedure, say pic.

3. Put the 'proc line' at the top (in the same input cell, using shift enter to open lines without executing the cell). The parameters of the problem are the input parameters of the procedure. In this case, these are a and b. 4. Put calculation of the answer in the body of the procedure, before the actual drawing. In this case, we calculate x and alpha in terms of a and b. 5. Put the 'end line' at the bottom. That's it. Execute the definition of the procedure, and test it.

Notes: Any variables which are assigned a value in the body of a procedure is considered to be a local variable (one whose value is only known to the procedure) by Maple. A warning is issued if that variable is not in the 'local line'. In the pic procedure below, all the local variables have been declared in the local line.

```
> pic := proc(a,b)
  local x,alpha,A,B,C,De,E,eps,trap,tri,labs;
> x := 2/3*a*3^(1/2)/(a^2/(b^2-b*a+a^2))^(1/2):
  alpha:=arcsin(1/2*3^(1/2)*(a^2/(b^2-b*a+a^2))^(1/2));
> A:= [0,0]:  B := [x*cos(alpha),0]:
  C:=[x*cos(alpha)+x*cos(2/3*Pi-alpha),0]:
  De:= [x*cos(alpha)+x*cos(2/3*Pi-alpha),b]:    E:= [0,a]:
> eps := .3:
> trap := plottools[polygon]([A,C,De,E],color=yellow):
> tri := plottools[polygon]([E,B,De],color=turquoise):
> labs := plots[textplot]([op(A-3*[0,eps]),"A"]),
  plots[textplot]([op(B-3*[0,eps]),"B"]),
  plots[textplot]([op(C-3*[0,eps]),"C"]),
> plots[textplot]([op(De+3*[eps,eps]),"D"]),
  plots[textplot]([op(E+3*[-eps,eps]),"E"]):
> plots[display](labs,tri,trap,scaling=constrained);
  end:
```

```
> pic(8,11);
```



After you get the drawing parameterized, you can add additional parameters, such as making the colors parameters, or the labels , or the offset values, or anything else you want.

These additional parameters can be added as **options** , similar to the way **ordinary plot options** can be added to a diagram. One must avoid using the same names as are used in the ordinary plot options, or you will get error messages. In the new procedure pic below, three options, Colors, Points, and Eps have been added.

```
> pic := proc(a,b)
  local x,alpha,A,B,C,De,E,eps,trap,tri,labs,colors, points,defaults,
  opts;
> defaults:=Colors=[yellow,turquoise],Points=["A","B","C","D","E"],Eps=.
  3;
  opts:= subs([
```

```
> op(select(type,[args],`=`)),defaults],[Colors,Points,Eps]);
  colors:=opts[1]:
  points:=opts[2]:
> eps:=opts[3]:
  x := 2/3*a*3^(1/2)/(a^2/(b^2-b*a+a^2))^(1/2):
> alpha:=arcsin(1/2*3^(1/2)*(a^2/(b^2-b*a+a^2))^(1/2));
  A:= [0,0]:  B := [x*cos(alpha),0]:
> C:=[x*cos(alpha)+x*cos(2/3*Pi-alpha),0]:
  De:= [x*cos(alpha)+x*cos(2/3*Pi-alpha),b]:    E:= [0,a]:

> trap := plottools[polygon]([A,C,De,E],color=colors[1]):
> tri := plottools[polygon]([E,B,De],color=colors[2]):
> labs := plots[textplot]([op(A-3*[0,eps]),points[1]]),
  plots[textplot]([op(B-3*[0,eps]),points[2]]),
  plots[textplot]([op(C-3*[0,eps]),points[3]]),
> plots[textplot]([op(De+3*[eps,eps]),points[4]]),
  plots[textplot]([op(E+3*[-eps,eps]),points[5]]):
> plots[display](labs,tri,trap,scaling=constrained,axes=none);
  end:

> pic(8,11,Colors=[red,green],Eps=0.1);
```



2.1.2 animations

If you have a parameterized drawing, then it is easy to make an animation which shows what happens when one of the numerical parameters is changed. The idea is to make a sequence of drawings with the parameter changing and then display (using plots[display]) these together with the option insequence=true chosen. For example, to see what happens as the right hand height of the trapezoid increases from 11 to 21 we could play the animation below.

```
> plots[display](seq(pic(8,11+i),i=0..10),insequence=true,scaling=const
  rained);
```



An interesting observation occurs. After the height gets past 16, the inscribed triangle is no longer inscribed! This is something that we might not have noticed had we not done the animation. (Initially, it seemed that perhaps we had written incorrect code for the animation. However, Kremer quickly asserted that the code was correct, because when the one height was more than twice the other, the side of the triangle is too large for the triangle to fit in the trapezoid. To verify this, go back and look at the algebra in the calculations and see that when CD $=$ a is more than twice AE $=$ b, then α is less than $\frac{\pi}{6}$, and so $\frac{2\pi}{3} - \alpha$ is greater than $\frac{\pi}{2}$ and so $\sin(\frac{2\pi}{3} - \alpha)$ is negative. So the segment AC $= x\sin(\alpha) + x\sin(\frac{2\pi}{3} - \alpha)$ is shorter than the segment AB $= x\sin(\alpha)$. So the equilateral triangle is not inscribed in that case.

```
> pic(8,16);
```



Now we can make an animation showing the inscribed equilateral triangle in each case (with a fixed AE).

```
> plots[display](seq(pic(20,10+i,Colors=[tan,blue]),i=0..30),seq(pic(20
  ,40-i),i=0..30),insequence=true,scaling=constrained);
```



2.1.3 A conjecture shot down

One might conjectured that the ratio of the area of the inscribed equilateral triangle to the area of the trapezoid remains constant. We can test that conjecture. From above,

```
> x := 2/3*a*3^(1/2)/(a^2/(b^2-b*a+a^2))^(1/2);
```

$$x := \frac{2\,a\,\sqrt{3}}{3\,\sqrt{\frac{a^2}{b^2 - b\,a + a^2}}}$$

```
> alpha:=arcsin(1/2*3^(1/2)*(a^2/(b^2-b*a+a^2))^(1/2));
```

$$\alpha := \arcsin\left(\frac{\sqrt{3}\,\sqrt{\frac{a^2}{b^2 - b\,a + a^2}}}{2}\right)$$

The the areas of the triangle and trapezoid are

```
> areatri := sqrt(3)/4*x^2;
```

$$areatri := \frac{\sqrt{3}\,(b^2 - b\,a + a^2)}{3}$$

```
> areatrap:= (a+b)/2*x*(cos(alpha)+cos(2/3*Pi-alpha));
```

$$areatrap := \frac{1}{3}\,\frac{(a+b)\,a\,\sqrt{3}\left(\frac{\sqrt{4 - \frac{3\,a^2}{b^2 - b\,a + a^2}}}{2} - \sin\left(\frac{\pi}{6} - \arcsin\left(\frac{\sqrt{3}\,\sqrt{\frac{a^2}{b^2 - b\,a + a^2}}}{2}\right)\right)\right)}{\sqrt{\frac{a^2}{b^2 - b\,a + a^2}}}$$

The ratio is

```
> rat := unapply(areatri/areatrap,a,b);
```

$$rat := (a, b) \rightarrow \frac{(b^2 - b\,a + a^2)\sqrt{\frac{a^2}{b^2 - b\,a + a^2}}}{(a+b)\,a\,(\frac{1}{2}\sqrt{4 - \frac{3\,a^2}{b^2 - b\,a + a^2}} - \sin(\frac{\pi}{6} - \arcsin(\frac{1}{2}\sqrt{3}\sqrt{\frac{a^2}{b^2 - b\,a + a^2}})))}$$

Doesn't look constant, but looks can be deceiving. Let's plot the ratio over its range.

```
> plot(rat(a,1),a=.5..2);
```



Ok, it's not constant. What we can say that the maximum ratio 2/3 occurs at the extremes when the trapezoid is a 60 90 trapezoid and the minimum ratio 1/2 occurs at the midpoint when the trapezoid is a rectangle.

3 Appendix for WHS authors

One should be able to come up with several interesting WHS problems based on this diagram. Here are a few. Note: We have set Latex_ equal to both. That enables tagit to produce latex and html versions of the problem. The latex versions can be extracted and latexed up separately. Here is a link ***Trapezoid_exercise.pdf***

The html versions are posted to WHS. To see them, log into WHS and add the class WHS authors.

```
> LATEX_:=both;
> tagit("The trapezoid ACDE shown below has an inscribed equilateral
  triangle EBD. Angle A and angle C are right angles.    If AE = 8 and CD
  = 11, what is the side of the
> triangle?",plots[display](pic(8,11,Colors=[magenta,green]),scaling=con
  strained,axes=none),"Answer =",_AC(2/3*291^(1/2)));
```

$$LATEX\_ := both$$

```
_LATEX

\item \

The trapezoid ACDE shown below has an inscribed equilateral triangle
EBD. Angle A and angle C are right angles.  If AE = 8 and CD = 11,
what is the side of the triangle?
```



$\!$

Answer =

 \rule{1in}{.1mm}

\vspace{12pt}

_KEY

\begin{tabular}{ll}

$\diamond$& $2/3\,\sqrt {291}$ \\

\end{tabular}

_YEK

QM_[0.05;2/3*291^(1/2)]

AH_[0]

The trapezoid ACDE shown below has an inscribed equilateral triangle EBD. Angle A and angle C are right angles. If AE = 8 and CD = 11, what is the side of the triangle?



Answer =

AC_[5]

SKIP_

```
> tagit("A rectangle has an inscribed equilateral triangle BDE, as
  shown below.    What is the ratio of the longest side to the shortest
  side?",plots[display](pic(8,8),scaling=constrained,axes=none),"Answer
  =",_AS([evalf(2/3*sqrt(3),5),1,1.2352]));
```

_LATEX

\item \

A rectangle has an inscribed equilateral triangle BDE, as shown below.
What is the ratio of the longest side to the shortest side?



$\!$

Answer =

\\ Circle correct answer:\\

\begin{tabular}{lll}

1 & $ 1.2352$ & $ 1.1547$ \\

\end{tabular}\\

\vspace{12pt}

_KEY

\begin{tabular}{ll}

$\diamond$& $ 1.1547$ \\

\end{tabular}

_YEK

QM_[0.05;1.1547]

AH_[0]

A rectangle has an inscribed equilateral triangle BDE, as shown below.
What is the ratio of the longest side to the shortest side?



Answer =

AS_[1.1547;1.2352;1]

SKIP_

```
> tagit("The trapezoid ACDE  has an inscribed equilateral triangle EDB.
  Angles A and C are right angles.  When is the ratio of the area of the
  triangle to the area of the trapezoid 2/3?","Answer:    When the
> trapezoid has an interior angle of ",_AS(["60 degrees","45 degree","30
  degree","70 degrees"]));
```

_LATEX

\item \

The trapezoid ACDE has an inscribed equilateral triangle EDB. Angles A and C are right angles. When is the ratio of the area of the triangle to the area of the trapezoid 2/3?

Answer: When the trapezoid has an interior angle of

\\ Circle correct answer:\\

\begin{tabular}{llll}

30 degree&60 degrees&45 degree&70 degrees\\

\end{tabular}\\

\vspace{12pt}

_KEY

\begin{tabular}{ll}

$\diamond$&60 degrees\\

\end{tabular}

_YEK

QM_[0.05;60 degrees]

AH_[0]

The trapezoid ACDE has an inscribed equilateral triangle EDB. Angles A and C are right angles. When is the ratio of the area of the triangle to the area of the trapezoid 2/3?

Answer: When the trapezoid has an interior angle of

AS_[70 degrees;60 degrees;30 degree;45 degree]

SKIP_

```
> tagit("When one side of the trapezoid gets too long relative the the
  other side, the equilateral triangle can no longer be inscribed in the
  interior of the trapezoid.      At what ratio does this occur ?",
> plots[display](seq(pic(8,12+i,Colors=[red,green]),i=0..10),insequence=
  true),
  "When the ratio of the long side to the short side =",
  _AC(2));
```

_LATEX

\item \

When one side of the trapezoid gets too long relative the the other side, the equilateral triangle can no longer be inscribed in the interior of the trapezoid.   At what ratio does this occur ?



$\!$

When the ratio of the long side to the short side =

\rule{1in}{.1mm}

\vspace{12pt}

_KEY

\begin{tabular}{ll}

$\diamond$& 2 \\

\end{tabular}

_YEK

QM_[0.05;2]

AH_[0]

When one side of the trapezoid gets too long relative the the other side, the equilateral triangle can no longer be inscribed in the interior of the trapezoid.   At what ratio does this occur ?



When the ratio of the long side to the short side =

AC_[5]

SKIP_